Pensieve Header: The universal Khovanov homology program described in UQAM, http://www.math.toronto.edu/-drorbn/Talks/UQAM-051001/index.html

Date: Tue, 27 Sep 2011 17:24:30 -0400 (EDT)
From: Dror Bar-Natan <drorbn@math.toronto.edu>
To: Scott Morrison <...>
Cc: Joshua Batson <...>
Subject: Re: formatting for UniversalKh

Double Hmmm,

Scary how little I remember. Anyway, you did remind me that I gave a talk about this once in UQAM, so I looked up my own handout and it came with a mathematica package, so now I more or less remember.

The handout is at http://www.math.toronto.edu/drorbn/Talks/UQAM-051001/4Tu-2.pdf. According to the “The work of Green” box at the bottom right (“work of Green” refers to the programming, not the math; the math is by Naot/DBN), the universal invariant is a komplex with objects formal arcs at various degrees (q^d) and homological heights (t^r), and with morphisms formal matrices of “curtains” with a number of handles (h^g) on each.

At the time I also wrote a short mathematica notebook to interpret the output of Green’s program. I’ve just edited it a tiny bit and re-posted it at http://katlas.math.toronto.edu/drorbn/AcademicPensieve/2005-10/. Reading that notebook you can get a hint for how to interpret the JavaKh output.

It seems that JavaKh only outputs the morphisms of the komplex; this is fair, because morphisms by definition carry the data on their domain and target objects. It seems that q^d*t^r*h^g*M[m,n,cs] stands roughly for:

  An m by n matrix of curtains with g handles with domain objects at degree d and height r; the entries of the matrix are given as a list of coefficients cs which still needs to be partitioned into rows.

What I wrote above is approximate; the precise thing is readable from the notebook and examples cited above.

Best,

Dror.

On Tue, 27 Sep 2011, Joshua Batson wrote:

> Hello Dror,
>
> I’m writing about universal mode in JavaKh. I’d like to compute the
> universal homology with F_2 coefficients, and scott recommended asking you
> for advice on parsing the output. It comes out like this (for the trefoil):
>
> q^-9*t^-3*h^0*M[0, 1] + q^-9*t^-3*h^1*M[1, 1, 1] + q^-9*t^-3*h^2*M[1, 1, 0]
> + q^-7*t^-3*h^0*M[1, 1, 0] + q^-7*t^-3*h^1*M[1, 1, 1] + q^-7*t^-2*h^0*M[0,
> 1] + q^-5*t^-2*h^0*M[0, 1] + q^-3*t^0*h^0*M[0, 1] + q^-1*t^0*h^0*M[0, 1]
>
> I’m not sure how to interpret the h and M[-,-,-] parts, and scott doesn’t
> remember either. We’re hoping that you do.
>
> Thanks,
>
> -Josh

```
$Path = $Path~Join~{"C:/drorbn/projects/KAtlas/"};
<< KnotTheory`
KhN[L_] := KhN[PD[L]];
KhN[pd_PD] := Module[
  {n, dir, f, cl, out},
  n = Max @@ (Max @@@ pd);
  pd1 = pd /. {
     X[n, i_, 1, j_] :> X[n, i, n + 1, j],
     X[i_, 1, j_, n] :> X[i, n + 1, j, n],
     X[1, j_, n, i_] :> X[n + 1, j, n, i],
     X[j_, n, i_, 1] :> X[j, n, i, n + 1]
    };
  dir = Directory[];
  SetDirectory[ToFileName[KnotTheoryDirectory[], "JavaKh"]];
  f = OpenWrite["pd", PageWidth → Infinity];
  WriteString[f, ToString[pd1]];
  Close[f];
  cl = StringJoin["!java -Xmx256m JavaKh -H < pd"];
  f = OpenRead[cl];
  out = Read[f, Expression];
  Close[f];
  SetDirectory[dir];
  out = StringReplace[out, {"q" → "#1", "t" → "#2"}];
  kh = ToExpression[out <> "&"][q, t];
  minr = Exponent[kh, t, Min];
  maxr = Exponent[kh, t, Max];
  obs = Expand[kh /. h → 0 /. M[_, n_, ___] :> Plus @@ Array[Arc, n]];
  obs = obs /. (q^j_.) * Arc[i_] :> Arc[j, i] /. Arc[i_] :> Arc[0, i];
  mos = Expand[
    h * kh /. {M[0, _] → 0, M[_, 0] → 0, h → H}
     /. M[m_, n_, cs___] :> Plus @@ Flatten[MapIndexed[
         (#1 * Curtain @@ Reverse[#2]) &,
         Partition[{cs}, n],
         {2}
        ]]
   ];
  mos = mos /. (q^j_.) * Curtain[k_, l_] :> Curtain[j, k, l] /. Curtain[k_, l_] :> Curtain[0, k, l];
  mos = mos /. (H^g_.) * Curtain[j_, k_, l_] :> H^(g - 1) Curtain[j, k, j + 2 (g - 1), l];
  Komplex @@ Table[{r, Coefficient[obs, t, r], Coefficient[mos, t, r]}, {r, minr, maxr}]
 ]
```

```
KhN[Knot[3, 1]]
```

```
Komplex[{-3, Arc[-8, 1], H Curtain[-8, 1, -6, 1]}, {-2, Arc[-6, 1], 0}, {-1, 0, 0}, {0, Arc[-2, 1], 0}]
```

```
Print /@ KhN[Knot[13, NonAlternating, 3663]];
```

{-6, Arc[-10, 1], H Curtain[-10, 1, -8, 1]}

{-5, Arc[-8, 1], 0}

{-4, Arc[-6, 1], -2 Curtain[-6, 1, -6, 1] - H Curtain[-6, 1, -4, 2]}

{-3, Arc[-6, 1] + Arc[-4, 1] + Arc[-4, 2],
H Curtain[-6, 1, -4, 1] + 2 Curtain[-4, 1, -4, 1] + H Curtain[-4, 1, -2, 1] - 2 Curtain[-4, 2, -4, 1]}

{-2, Arc[-4, 1] + Arc[-4, 2] + Arc[-2, 1], H Curtain[-4, 2, -2, 2]}

{-1, Arc[-2, 1] + Arc[-2, 2] + Arc[0, 1],
H Curtain[-2, 1, 0, 1] - 2 Curtain[0, 1, 0, 1] + 2 Curtain[0, 1, 0, 2] + H Curtain[0, 1, 2, 1]}

{0, Arc[0, 1] + Arc[0, 2] + Arc[0, 3] + Arc[2, 1], H Curtain[0, 2, 2, 1] + H Curtain[0, 2, 2, 2] -
2 H Curtain[0, 3, 2, 1] - 2 H Curtain[0, 3, 2, 2] - 2 Curtain[2, 1, 2, 1] - 2 Curtain[2, 1, 2, 2]}

{1, Arc[0, 1] + Arc[2, 1] + Arc[2, 2], H Curtain[0, 1, 2, 1] - H Curtain[2, 1, 4, 2] + H Curtain[2, 2, 4, 2]}

{2, Arc[2, 1] + Arc[4, 1] + Arc[4, 2], H Curtain[4, 1, 6, 1]}

{3, Arc[4, 1] + Arc[6, 1], H Curtain[4, 1, 6, 2]}

{4, Arc[6, 1] + Arc[6, 2], H Curtain[6, 1, 8, 1]}

{5, Arc[8, 1], 0}

{6, Arc[10, 1], H Curtain[10, 1, 12, 1]}

{7, Arc[12, 1], 0}

KhN[TorusKnot[6, 5]]

Komplex[{0, Arc[20, 1], 0}, {1, 0, 0}, {2, Arc[24, 1], H Curtain[24, 1, 26, 1]},
{3, Arc[26, 1], 0}, {4, Arc[26, 1], -H^{2} Curtain[26, 1, 30, 1]},
{5, Arc[30, 1], 0}, {6, Arc[28, 1] + Arc[30, 1],
H Curtain[28, 1, 30, 1] - 2 Curtain[30, 1, 30, 1] - H Curtain[30, 1, 32, 1]},
{7, Arc[30, 1] + Arc[32, 1], 0}, {8, Arc[30, 1] + Arc[32, 1],
2 H^{2} Curtain[30, 1, 34, 1] + H^{3} Curtain[30, 1, 36, 1] + 5 H Curtain[32, 1, 34, 1] +
H Curtain[32, 1, 34, 2]}, {9, Arc[34, 1] + Arc[34, 2] + Arc[36, 1],
-H Curtain[34, 1, 36, 1] + 5 H Curtain[34, 2, 36, 1] + 2 Curtain[36, 1, 36, 1]},
{10, Arc[34, 1] + Arc[36, 1], 5 H Curtain[34, 1, 36, 1] + H^{2} Curtain[34, 1, 38, 1]},
{11, Arc[36, 1] + Arc[38, 1], H^{2} Curtain[36, 1, 40, 1] - 5 H Curtain[38, 1, 40, 1]},
{12, Arc[36, 1] + Arc[40, 1], 3 H^{2} Curtain[36, 1, 40, 1] + H^{3} Curtain[36, 1, 42, 1]},
{13, Arc[40, 1] + Arc[42, 1], H Curtain[40, 1, 42, 1] - 3 Curtain[42, 1, 42, 1]},
{14, Arc[42, 1], 0}]